

Maple Tutorial #3: Derivatives and Integrals*

Scott R. Fulton and Christos A. Xenophontos
Department of Mathematics and Computer Science

Clarkson University

Maple V Release 5
Fall 1999

Introduction

After working through the first two tutorials, you should be able to use Maple to do much of pre-college mathematics, including arithmetic, algebraic manipulations, solving equations, and graphing functions. This is a good start towards being able to use Maple as a tool to explore mathematics symbolically, numerically, and graphically.

However, Maple can do mathematics far beyond what you did in high school. In this third tutorial, we'll learn how to use Maple to attack some college-level topics. The goal is to learn how to:

- Compute derivatives
- Evaluate integrals exactly
- Approximate integrals numerically

As before, you'll get the most out of this tutorial if you read the text carefully and *try* the commands as shown. After finishing this tutorial, you should know enough to begin using Maple as a tool to explore and apply calculus.

*Technical Report 99-05, Department of Mathematics and Computer Science, Clarkson University, Potsdam, NY 13699-5815.

Derivatives

You can use Maple to compute derivatives analytically (that is, using formulas—not numbers). The command you use depends on how you choose to represent the function: as an *expression* or as a *function*. Note: If you're not sure of the distinction between expressions and functions in Maple, review Tutorial #2 (pages 1-2) before proceeding.

To differentiate expressions, use “diff”:

Suppose you want to take the derivative of $y = x^5$. To define this as an expression, you would type:

```
> y := x^5;
```

$$y := x^5$$

To differentiate this (that is, to find its derivative), type:

```
> diff( y, x );
```

$$5x^4$$

Note that the result comes back as an expression: `diff` takes an expression as its first input argument, and returns the derivative of that expression as another expression. Of course, you could combine the above two commands as

```
> diff( x^5, x );
```

$$5x^4$$

Note that `diff` requires a second argument (in the above example, this is `x`). This argument tells Maple what the independent variable is. Why is this needed? Suppose you want to differentiate $y = c * x^5$, where c is a constant. You tell Maple that x is the variable by typing:

```
> diff( c*x^5, x );
```

$$5cx^4$$

Note that if x were a constant and c the variable, you could type:

```
> diff( c*x^5, c );
```

$$x^5$$

to find the derivative of $c * x^5$ with respect to c (not x).

You can also compute higher-order derivatives by repeating the independent variable. For example, second and third derivatives are computed as

```
> diff( c*x^5, x, x );
```

$$20cx^3$$

and

```
> diff( c*x^5, x, x, x );
```

$$60 c x^2$$

Alternatively, you could use the \$ sign to indicate the number of differentiations you would like to perform. For example, to obtain the third derivative you type

```
> diff(c*x^5,x$3)
```

$$60 c x^2$$

To differentiate functions, use “D”:

Often, it’s more convenient to use functions (rather than expressions) in Maple: they take a bit more typing, but it’s easier to plug numbers into them. For example, we could represent the function $y = f(x) = x^5$ as a function in Maple by typing

```
> f := x -> x^5;
```

$$f := x \rightarrow x^5$$

This allows us to evaluate it by plugging numbers in directly:

```
> f(2); f(3.5); f(-7.14);
```

$$32$$

$$525.21875$$

$$-18556.28606$$

To find the derivative of a function f , use the Maple command D:

```
> D(f);
```

$$x \rightarrow 5 x^4$$

Note that the result comes back as a function (not a expression—see the arrow?): D takes a function as input and returns its derivative as a function. In fact, it’s often convenient to give this derivative function a name, like:

```
> fprime := D(f);
```

$$fprime := x \rightarrow 5 x^4$$

for then we can do things like evaluate it:

```
> fprime(2); fprime(3.5); fprime(-7.14);
```

$$80$$

$$750.3125$$

$$12994.59808$$

or plot it:

```
> plot( { f(x), fprime(x) }, x=-3..3 );
```

Of course, if f is a function then $f(x)$ is an expression (the output of the function f at x) so we could use `diff` on that:

```
> diff( f(x), x );
```

$$5x^4$$

However, we cannot say:

```
> D(f(x));
```

$$5D(x)x^4$$

or:

```
> diff( f, x );
```

$$0$$

Do you see *why* both of these are wrong?

A Cautionary Note:

If you ask for the derivative of something you haven't yet defined, Maple will represent it using a notation you may not expect. For example, try typing:

```
> diff( g(x), x );
```

$$\frac{\partial}{\partial x} g(x)$$

The result you get (if you haven't defined a function g yet) uses the symbol “ ∂ ”. This is the symbol for a “partial derivative”, which is used for functions of several variables (as in Calculus III). Since Maple doesn't know how many variables your function may use, it tries to play it safe and use a general notation. This can be confusing if you've never seen it before, but it's not a big deal.

Integrals

Maple can compute both indefinite and definite integrals. For Maple's integration commands you express the integrand as an *expression*, rather than a *function*—if you're not sure of the difference, review Tutorial #2 (pages 1-2) before proceeding. However, you'll often want to do other things with the integrand (such as evaluate it at some point or plot it to see what's going on), so it may be convenient to define it as a function. For example, if we're going to integrate and otherwise work with the “square” function, we would define it as

```
> f := t -> t^2;
```

$$f := t \rightarrow t^2$$

The output of that function is then given by the expression $f(t)$; which is what you will input to Maple's integration (or plotting) commands.

Indefinite Integrals:

To compute the indefinite integral (antiderivative), you use the command `int`, which takes two arguments: an *expression* for the integrand, and the variable of integration. Since here we've defined `f` as a function, $f(t)$ is an expression giving the output of that function, so we can compute $\int t^2 dt$ by typing:

```
> int( f(t), t );
```

$$\frac{1}{3}t^3$$

Note that Maple doesn't add the constant of integration. You could check this result by differentiating it:

```
> diff( %, t );
```

$$t^2$$

Notice that the result is the same as the original integrand.

Definite Integrals:

To evaluate a definite integral, you could find the antiderivative as above, and then evaluate it at both endpoints and subtract (thus using the Fundamental Theorem of Calculus). Since this is such a common calculation, Maple has it built in: all you need to do is add the limits to the command `int`. For example, to compute $\int_0^2 t^2 dt$ you would type:

```
> int( f(t), t=0..2 );
```

$$\frac{8}{3}$$

Of course, if you are just going to evaluate a single integral, you don't need to define a Maple function `f` for the integrand; you could simply type

```
> int( t^2, t=0..2 );
```

$$\frac{8}{3}$$

Numerical Values of Integrals:

Your calculus instructor may not have mentioned it, but many functions simply don't have antiderivatives (indefinite integrals) which can be expressed in terms of simple functions.

Maple knows about the most common such cases, and expresses the result—when it can—using other (more obscure) functions. For example,

```
> int( sin(t^2), t=0..2 );
```

$$\frac{1}{2} \text{FresnelS} \left(2 \frac{\sqrt{2}}{\sqrt{\pi}} \right) \sqrt{2} \sqrt{\pi}$$

Even if you've never heard of the function “FresnelS”, it probably doesn't matter: if all you want is the numerical value, you can get it by

```
> evalf(%);
```

```
.8047764895
```

An even easier way to “request” a numerical value is to use floating-point values for the limits of integration. For example,

```
> int( sin(t^2), t=0.0..2.0 );
```

```
.8047764895
```

(note we've changed 0 to 0.0 and 2 to 2.0).

In other cases where there is no antiderivative in terms of standard functions, Maple still tries to evaluate a definite integral exactly (that is, without using floating-point approximations). Sometimes it works a long time before getting the answer; sometimes it just gives up, in which case it simply returns the integral unevaluated:

```
> int( sin(t^3), t=0..2 );
```

$$\int_0^2 \sin(t^3) dt$$

If your computer doesn't have enough memory, Maple may take even longer (minutes...or hours...or days...) or possibly crash the machine. In all of these cases, if what you're after is the numerical value of the integral, it's better to simply tell Maple to evaluate it numerically from the outset. A natural thing to try is

```
> evalf( int( sin(t^3), t=0..2 ) );
```

```
.4519484772
```

Indeed, this works, but at a price: Maple first tries to evaluate the integral exactly, and only turns to the numerical approximation *after* it has wasted all that time trying to find an antiderivative [if you're typing these commands in sequence here, you may not notice this effect, since Maple “remembers” that it couldn't do this integral]. To tell Maple to simply set up the integral, skip trying to find the antiderivative, and go directly to numerical approximation, use

```
> evalf( Int( sin(t^3), t=0..2 ) );
```

```
.4519484772
```

Here, the command `Int` (with the upper-case `I`) is the “inert” version of the integration command `int`: it tells Maple to set up the integral but not evaluate it (which is left to `evalf` to accomplish). Note that you can use this approach to construct antiderivatives for functions where the antiderivatives aren’t known explicitly: for example, you can define

```
> g := t -> sin(t^3);   G := x -> evalf( Int( g(t), t=0..x ) );
```

$$g := t \rightarrow \sin(t^3)$$

$$G := x \rightarrow \text{evalf} \left(\int_0^x g(t) dt \right)$$

and then plot the function `g` and its antiderivative `G` on the same graph with

```
> plot( {g(x), G(x)}, x=0..2 );
```

Simple Numerical Integration Rules

If you want an accurate approximation to a difficult integral, you should use the commands `evalf` and `Int` as described above: this approach uses Maple’s built-in numerical integration scheme, which is reasonably sophisticated. Maple also has commands for the more basic numerical integrations methods usually studied in Freshman Calculus (the rectangle, trapezoid, and Simpson rules). Since these commands are intended primarily to help you learn—rather than solve problems—they are not automatically read into Maple each time, but located in a separate package (called the “student” package). To read this package into Maple you use

```
> with(student):
```

Note that the colon at the end here suppresses the long list of new commands being read in; if you want to see them, use a semicolon instead. Once you’ve loaded this package, it stays there throughout your current Maple session—you don’t need to load it again.

The simplest numerical approximations to definite integrals are the left and right sums (or rectangle rules). Suppose you want to compute the left sum approximation to $\int_0^2 t^2 dt$. If you didn’t do so before, define the integrand as the function

```
> f := t -> t^2;
```

You can plot the integrand over the interval $[0, 2]$ with the command

```
> plot( f(t), t=0..2 );
```

Similarly, the command `leftbox` draws a picture of the leftsum approximation; it uses the same input as the `plot` command, plus a number at the end to tell Maple how many intervals (or boxes) to use:

```
> leftbox( f(t), t=0..2, 8 );
```

(here we’re using 8 boxes). To evaluate the left sum (the sum of the areas of the boxes you just drew) use the command `leftsum`, which has the same input arguments as `leftbox` did:

```
> leftsum( f(t), t=0..2, 8 );
```

$$\frac{1}{4} \left(\sum_{i=0}^7 \left(\frac{1}{16} i^2 \right) \right)$$

Note that Maple's output is an unevaluated sum (Maple does this so you can see the details—remember, this is an "educational" command, not an "operational" command). To evaluate this sum, you have two choices: to get a decimal (floating-point) value, use `evalf`; to get an exact value (if possible), use `value`. You can combine the commands as follows:

```
> evalf( leftsum( f(t), t=0..2, 8 ) );
```

$$2.187500000$$

or

```
> value( leftsum( f(t), t=0..2, 8 ) );
```

$$\frac{35}{16}$$

Note that there is no real need to draw the boxes if you just want to compute the left sum approximation—but seeing them may help you understand what the approximation means.

Maple has similar commands `rightsum`, `middlesum`, `trapezoid`, and `simpson` for the analogous rules (only the first two have corresponding graphical representations, namely `rightbox` and `middlebox`). These work in exactly the same way as `leftsum`. The only tricky point is the definition of the number of intervals in `simpson`: some texts define the "n" for Simpson's rule as the number of subintervals over which the basic 3-point Simpson rule is applied, whereas the Maple command `simpson` takes twice that value, thus specifying the number of integration points.

If you're going to experiment with numerical integration commands, it may be helpful to define new functions which apply them as follows:

```
> g := t -> 1 + 2*t - t^2/4;
> a := 0; b := 4;
> left := n -> evalf( leftsum(g(t),t=a..b,n) );
> trap := n -> evalf( trapezoid(g(t),t=a..b,n) );
> simp := n -> evalf( simpson(g(t),t=a..b,n) );
```

For example, `trap` is a function: its input is the number of subintervals to use, and its output is the value computed by the trapezoid rule. To use these you can type:

```
> left(10); trap(10); simp(10);
> left(20); trap(20); simp(20);
```

and so forth. You could even do something like

```
> for k from 1 to 8 do print( 2^k, left(2^k), trap(2^k), simp(2^k) ) od;
```

to compare the convergence of these methods. What do you notice for this particular integral? Do you get the same behavior for the integrand t^3 ? How about t^4 ?

Can I trust Maple to get the right answer every time?

In a word—no. Maple almost always gets the right answer—or gives up—but occasionally it gets fooled. For example, Release 3 of Maple had trouble with

```
> evalf( int( sqrt( x^4 + x^2 ), x=-3..-1 ) );
```

returning a negative value, even though the integrand is positive. Curiously enough, replacing `int` with `Int` here switched the sign—can you guess why? This example works properly in Releases 4 and 5, but there's no guarantee there aren't other bugs. Even though Maple makes errors only rarely, it's a good idea to think about whether the answers it gives make sense, and to check them where possible. Plotting pieces of the problem is often an excellent way to see what's going on, and checking that the answers are reasonable.

Unfortunately, Maple cannot correct your errors. One student typed

```
> int( x*lnx, x );
```

and then checked the answer by differentiating:

```
> diff( %, x );
```

Since this result matched the integrand, the student assumed that the integral was correct. However, it wasn't—can you see why? (Hint: the natural logarithm is a function).

Moral: Know what you're doing—be careful—check everything.

Some Maple Commands for Calculus

Command	Description
<code>diff(y, x);</code>	differentiate the <i>expression</i> <code>y</code> with respect to <code>x</code>
<code>D(f);</code>	differentiate the <i>function</i> <code>f</code> of a single variable
<code>int(f(x), x);</code>	compute the indefinite integral $\int f(x) dx$
<code>int(f(x), x=a..b);</code>	compute the definite integral $\int_a^b f(x) dx$
<code>Int(f(x), x=a..b);</code>	set up (but do not evaluate) an integral
<code>leftbox(f(x), x=a..b, n);</code>	draw a left sum approximation
<code>leftsum(f(x), x=a..b, n);</code>	left sum approximation to $\int_a^b f(x) dx$
<code>rightsum(f(x), x=a..b, n);</code>	right sum approximation to $\int_a^b f(x) dx$
<code>middlesum(f(x), x=a..b, n);</code>	middle sum approximation to $\int_a^b f(x) dx$
<code>trapezoid(f(x), x=a..b, n);</code>	trapezoidal rule approximation to $\int_a^b f(x) dx$
<code>simpson(f(x), x=a..b, n);</code>	Simpson's rule approximation to $\int_a^b f(x) dx$